
1. Figure S1. 

(A) Shorter variant of BANGO SEA XLOC‒H2OBC pulse sequence

Narrow filled, wider filled and open bars refer to /4, /2 and  pulses, respectively. A1 is adiabatic inversion, CAWURST-20 pulse (0.5 ms, 300 ppm, 
low-to-high frequency sweep). The BANGO delay  is set to (1Jmax + 1Jmin)-1. A 2nd order low-pass J filter is employed with 1 = 0.5[1Jmin + 0.146 (1Jmax ‒ 
1Jmin)]-1 and 2 = 0.5[1Jmax ‒ 0.146 (1Jmax ‒ 1Jmin)]-1. ε = t(πH)/2. The delays Δ and T are for evolution under heteronuclear long-range and homonuclear 
proton-proton couplings, respectively.  a gradient delay. The amplitude of the purging gradients (open triangle) can be set an order of magnitude lower 
than the amplitude of the other ones selecting coherence transfer echo and antiecho. 1/3 = {x, -x, -x, x}, 2/4 = {x, x, -x, -x}, and receiver phases 1/2 
= {x,-x}.

(B) BANGO HMBC‒H2OBC pulse sequence 

Narrow filled, wider filled and open bars refer to /4, /2 and  pulses, respectively. A0, A1 and A2 are adiabatic refocussing and inversion pulses, 
composite 20% smoothed CHIRP (2 ms, 80 kHz sweep; Crp80comp.4) and CAWURST-20 (0.5 ms, 300 ppm, low-to-high frequency sweep), respectively. 
BANGO delay  is set to (1Jmax + 1Jmin)-1. A 2nd order low-pass J filter is employed with 1 = 0.5(1Jmin)-1 and 2 = 0.5(1Jmax)-1. ’ =  - t(A2)/2. ε = t(πH)/2. Δ 
and T delays are for evolution under heteronuclear long-range and homonuclear proton-proton couplings, respectively.  a gradient delay. The amplitude 
of the four purging gradients (open triangles) can be set an order of magnitude lower than the amplitude of the other ones selecting coherence transfer 
echo or antiecho. 0 = x, 1/4 = {x, -x, -x, x}, 2/5 = {x, x, -x, -x}, 3 = x, and receiver phases 1/2 = {x,-x}. Before standard processing the obtained 
combined data set is separated into two blocks, corresponding to HMBC and H2OBC data, using the Bruker au-program splitx. Then the HMBC and 
H2OBC data are processed as in their standalone experiments. 

(C) Shorter variant of BANGO HMBC‒H2OBC pulse sequence

Narrow filled, wider filled and open bars refer to /4, /2 and  pulses, respectively. A0 and A1 are adiabatic refocussing and inversion pulses, composite 
20% smoothed CHIRP (2 ms, 80 kHz sweep; Crp80comp.4) and CAWURST-20 (0.5 ms, 300 ppm, low-to-high frequency sweep), respectively. BANGO 
delay  is set to (1Jmax + 1Jmin)-1. A 2nd order low-pass J filter is employed with 1 = 0.5(1Jmin)-1 and 2 = 0.5(1Jmax)-1. ε = t(πH)/2. Δ and T delays are for 
evolution under heteronuclear long-range and homonuclear proton-proton couplings, respectively.  a gradient delay. The amplitude of the four purging 
gradients (open triangles) can be set an order of magnitude lower than the amplitude of the other ones selecting coherence transfer echo or antiecho. 0 
= x, 1/4 = {x, -x, -x, x}, 2/5 = {x, x, -x, -x}, 3 = x, and receiver phases 1/2 = {x,-x}. 

2. Figure S2. Excerpt of HMBC spectra of ibuprofen recorded with the experiment in Fig. S1 (red) and with 

standolane HMBC experiment (black)

700 MHz HMBC spectrum (red) of ibuprofen (0.5 M in CDCl3) recorded on a Bruker Avance NEO spectrometer equipped with a TCI z-gradient prodigy 
probe using the pulse sequence in Fig. S1. HMBC spectrum (black) recorded in a standalone experiment is displaced along F1 for better visualization. 
The F2 sections next to the contour plot extracted at C1, C2/C9 and C10 from the red and black spectra, respectively, demonstrate the improved sensitivity 
of the combined experiment. The averaged improvement in sensitivity is about 6%. The spectra were acquired with the parameters: Δ = 83 ms, T = 23 
ms, 1Jmin = 125 Hz and 1Jmax = 165 Hz, spectral widths of 8.4 ppm (1H) and 190.0 ppm (13C), relaxation delay 1.7 s, 512 points in t1 with 1 scan per 
increment and 2048 data points in t2. 

3. BANGO SEA XLOC‒H2OBC pulse sequence code for Bruker spectrometers                                      
(Avance II, III and NEO systems)

;bango_sea-xloc_h2obc_ek_ows
;avance-version – tested on Avance II and NEO systems, TopSpin 2.1 and TopSpin 4.0.2

;SEA XLOC
;2D H-1/X correlation via heteronuclear zero (ZQ) or double quantum (DQ) coherence
;recorded in two separate (SEA = separate echo-antiecho) experiments 
;optimized on long range couplings
;with second order low-pass J-filter to suppress one-bond correlations
;no decoupling during acquisition
;using gradient pulses for coherence selection
;using hard or shaped pulses for inversion on f2 - channel

;T. Gyongyosi, T.M. Nagy, K.E. Kover & O.W. Soerensen,
; Chem. Commun. 54, 9781-9784 (2018)

;H2OBC 
;2D H-H & H-1/X correlation via heteronuclear zero and double quantum 
;coherence with carbon multiplicity editing
;phase sensitive using Echo/Antiecho gradient selection 

;E. Kupce & O.W. Soerensen, 
; Magn. Reson. Chem. 55, 515-518 (2017) 
;N.T. Nyberg, J.O. Duus & O.W. Soerensen, 
; J. Am. Chem. Soc. 127, 6154-6155 (2005)

;May 21, 2019  KEK, TGY, TMN

;This pulse sequence is part of
;Tamás Milán Nagy, Tamás Gyöngyösi, Katalin E. Kövér, Ole W. Sørensen; " BANGO SEA XLOC/HMBC 
;–H2OBC: Complete heteronuclear correlation within minutes from one NMR pulse sequence”

;manuscript in preparation

;The pulse sequence has been coded for test purposes only and 
;may contain errors. 
;The functionality of the pulse sequence itself may differ depending on
;the hardware as well as the software used to execute it. Functionality 
;on differing systems cannot be granted.
;Any use of this pulse sequence on a spectrometer is at your own risk.
;
;By using this pulse sequence, or any modification of it in any published material
;you agree to acknowledge the above-mentioned publication.

;Combined acqusition of two experiments = SEA XLOC (magnitude) and 2BOB (phase sensitive, 
;echo-antiecho)

;set FNmode - UNDEFINED and NBL=2

;PROCESSING:

;FIRST separate the two datasets using AU splitx
;Second data set corresponds to H2OBC and can be processed with xfb (set echo-antiecho, pk)
;Separate the first data set with AU split into DQ and ZQ data - then process both 
;with xfb (set QF, mc in F1)

;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"cnst30=sfo1/(sfo2*4)"

define list<gradient> EA1 = {cnst30 -cnst30}
define list<gradient> EA2 = {50 30}  ;DQ- and ZQ-selection in SEA XLOC 

"p2=p1*2"
"p4=p3*2"

"d0=3u"
"d11=30m"

"TAU=1s/(cnst6 + cnst7)"
"TAU1=d21/2-TAU*2-p14-d0-p2/2"  ;d21: constant time for evolution of J(HH) [16 - 23 msec]
"d22=TAU1"
"TAU4=TAU-p16"
"TAU8=TAU4-d0*2-p2"
"d29=TAU1-8u"

"in0=inf1/2"
"in19=in0"
"in22=in0"
"in29=in0"

"d23=1s/(cnst14*4)"        ;cnst14 – heteronuclear long-range coupling constant
"d19=d23*2-p16*2-d16*2-d0" ;long-range coupling evolution delay  

"td1=tdmax(td1,d29,in29)"

"l3 = td1/4"    ;NBL = 2, data from two experiments are stored in one file !!

"DELTA1=1s/(2 * (cnst6 + 0.07 * (cnst7-cnst6) ) )"
"DELTA3=1s/(2 * (cnst7 - 0.07 * (cnst7-cnst6) ) )"
"DELTA4 = d16 + p16"
"DELTA2=(DELTA1+DELTA3)/2-DELTA4"
"DELTA5=DELTA2-p30/2"
"DELTA8=DELTA2+p30/2-8u"

"DELTA9=1s/(2 * (cnst6 + 0.146 * (cnst7-cnst6)) ) -p16-d16"
"DELTA12=1s/(2 * (cnst7 - 0.146 * (cnst7-cnst6)) ) -p16-d16"

"d2=1s/(cnst2*2)"      ;cnst2 = 145

1 ze
2 d11
3 d1 do:f2 st0       ;NBL = 2 

  (p1*0.5 ph1):f1    ;SEA XLOC starts here with BANGO using 45 degree 1H pulse
  d2 pl2:f2
  (center (p2 ph1) (p4 ph1):f2 )
  d2
  (p1*0.5 ph1):f1

  DELTA9 UNBLKGRAD ;second order low-pass J-filter to suppress one-bond correlations
  p16:gp4                     
  d16 pl2:f2
  (p3 ph1):f2
  DELTA12
  p16:gp5
  d16

  (p3 ph1):f2
  p16:gp6
  d16 pl2:f2
     ;heteronuclear long-range coupling evolution starts
  d0                                  ;d0 incremented
  (center (p2 ph1) (p4 ph1):f2 )
  d19                                 ;d19 decremented
  p16:gp11*EA2                        ;gpz11 = 1 multiplied by 50 for DQ-selection 
  d16                                 ;by 30 for ZQ-selection
 
  (p3 ph14):f2
  p16:gp2
  d16 BLKGRAD
  
  d0
  d0

  (p3 ph15):f2

  goscnp ph30

;second experiment starts here: H2OBC

22 d11 st            ;
23 d11 do:f2         

d11 pl12:f2
50u UNBLKGRAD

p16:gp12             ;weak purging gradient between the two experiments 
d16

20u cpd2:f2
(p1 ph1)
d22                 ;d22 decremented
TAU do:f2
4u pl2:f2
(p3 ph3):f2
TAU pl0:f2
(p14:sp30 ph1):f2   ;adiabatic CA-WURST inversion pulse

      ;sp30:wvm:ad180Cj2: cawurst-20(300 ppm, 0.5 ms; L2H) 

d0
(p2 ph1)
d0
p16:gp1*EA1
TAU8
d25
4u
(p14:sp30 ph1):f2
4u pl2:f2
(p3 ph5):f2
p16:gp1*EA1*-1
TAU4 pl12:f2
d29 cpd2:f2        ;d29 decremented
4u do:f2
4u pl2:f2

(p1 ph2) (p3 ph1):f2

p16:gp1*0.5 
d16 
DELTA5 pl0:f2 
                    ;adiabatic CA-WURST inversion pulse
(p30:sp17 ph1):f2   ;sp17:wvm:ad180Cref: cawurst-20(300 ppm, 0.5 ms)

(p2 ph1) 
 
p16:gp1*0.75 
d16 
DELTA8 pl12:f2 
4u cpd2:f2 
4u BLKGRAD

go=2 ph31 
                            
d11*0.5 do:f2 wr #0 if #0 zd    
d11*0.5 igrad EA1
d11*0.5 igrad EA2

lo to 2 times 2

d11*0.25 id0 
d11*0.25 dd19
d11*0.25 dd22 
d11*0.25 dd29 
d11*0.25 ip3*2 
d11*0.25 ip31*2

lo to 2 times l3    ; l3 = td1/4   NBL = 2 !

exit

ph1=0 
ph14=0 2 2 0
ph15=0 0 2 2

ph2=1 
ph3=0 2 2 0 
ph5=0 0 2 2 
ph6=0 

ph30=0 2
ph31=0 2  

;pl1 : f1 channel - power level for pulse (default) 
;pl2 : f2 channel - power level for pulse (default) 
;pl12: f2 channel - power level for CPD/BB decoupling 
;p1 : f1 channel - 90 degree high power pulse 
;p2 : f1 channel - 180 degree high power pulse 
;p3 : f2 channel - 90 degree high power pulse 
;p16: homospoil/gradient pulse [1 msec] 
;p24: f2 channel - 180 degree shaped pulse for refocussing 
; = 2msec for Crp80comp.4 
;d0 : incremented delay (2D) [3 usec] 
;d1 : relaxation delay; 1-5 * T1 
;d11: delay for disk I/O [30 msec] 
;d16: delay for homospoil/gradient recovery 
;d22: decremented delay (2D)
;d21: constant time (evolution of J(HH)) [16 - 23 msec] 
;d29: decremented delay (2D)
;cnst6: = 1J(XH)min 
;cnst7: = 1J(XH)max 
;inf1: 1/SW(X) = 2 * DW(X) 
;in0: 1/(2 * SW(X)) = DW(X) 
;in19: = in0 
;in22: = in0 
;in29: = in0 
;nd0: 2 
;ns: 2 * n 
;ds: 16 
;td1: number of experiments 

;FnMODE: UNDEFINED

;cpd2: decoupling according to sequence defined by cpdprg2 
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence 

; ~~~~~~~~~~~~~~ WaveMaker Shapes ~~~~~~~~~~~~~~~~~~~~~~~~~~ 
;USE: wvm  command in TopSpin command line to generate CA-WURST adiabatic shape pulses
;sp30:wvm:ad180Cj2: cawurst-20(300 ppm, 0.5 ms; L2H) 
;sp17:wvm:ad180Cref: cawurst-20(300 ppm, 0.5 ms)

;for z-only gradients: 

;use gradient files: 
;gpnam…: SMSQ10.100 

;gpz1 = 80%
;gpz2: -40.1% for C-13
;gpz3:  40.1%
;gpz4: 60%
;gpz5: -40%
;gpz6: -20%
;gpz11: 1 multiplied by 50% for DQ or 30% for ZQ for C-13
;gpz12 = 13% purging gradient 

4. BANGO HMBC‒H2OBC pulse sequence code for Bruker spectrometers                                            

(Avance II, III and NEO systems)
;bango_hmbc_h2obc_ek_ows
;avance-version – tested on Avance II and NEO systems, TopSpin 2.1 and TopSpin 4.0.2

;HMBC
;2D H-1/X correlation via heteronuclear zero (ZQ) or double quantum (DQ) coherence
;phase sensitive using Echo/Antiecho gradient selection
;with second order low-pass J-filter to suppress one-bond correlations
;no decoupling during acquisition

;H2OBC 
;2D H-H & H-1/X correlation via heteronuclear zero and double quantum 
;coherence with carbon multiplicity editing 
;phase sensitive using Echo/Antiecho gradient selection 

;E. Kupce & O.W. Soerensen, 
; Magn. Reson. Chem. 55, 515-518 (2017) 
;N.T. Nyberg, J.O. Duus & O.W. Soerensen, 
; J. Am. Chem. Soc. 127, 6154-6155 (2005)

;May 21, 2019  KEK, TGY, TMN

;This pulse sequence is part of
;Tamás Milán Nagy, Tamás Gyöngyösi, Katalin E. Kövér, Ole W. Sørensen; " BANGO SEA XLOC/HMBC 
;–H2OBC: Complete heteronuclear correlation within minutes from one NMR pulse sequence”

;manuscript in preparation

;The pulse sequence has been coded for test purposes only and 
;may contain errors. 
;The functionality of the pulse sequence itself may differ depending on
;the hardware as well as the software used to execute it. Functionality 
;on differing systems cannot be granted.
;Any use of this pulse sequence on a spectrometer is at your own risk.
;
;By using this pulse sequence, or any modification of it in any published material

;you agree to acknowledge the above-mentioned publication.

;Combined acqusition of two experiments = HMBC and 2BOB (phase sensitive, echo-antiecho)

;set FNmode – echo-antiecho and NBL=2

;PROCESSING:

;FIRST separate the two datasets using AU splitx
;then both datasets can be processed with xfb (echo-antiecho, pk), 
;for hmbc use xf2m for magnitude calculatoin in F2

#include <Avance.incl>
#include <Grad.incl>
#include <Delayek.incl>

"cnst30=sfo1/(sfo2*4)"
"cnst31=(1-sfo2/sfo1)/(1+sfo2/sfo1)"

define list<gradient> EA1 = {cnst30 -cnst30}
define list<gradient> EA3 = { 1.000 -cnst31}    ; HMBC phase sensitive, echo-anti echo
define list<gradient> EA4 = { -cnst31 1.000}

"p2=p1*2"
"p4=p3*2"

"d0=3u"
"d11=30m"

"TAU=1s/(cnst6 + cnst7)"
"TAU1=d21/2-TAU*2-p14-d0-p2/2"
"d22=TAU1"
"TAU4=TAU-p16"
"TAU8=TAU4-d0*2-p2"

"d29=TAU1-8u"

"in0=inf1/2"
"in22=in0"
"in29=in0"

"d6=1s/(cnst14*2)"     ;cnst14 = 6 for HMBC long-range J(CH)

"td1=tdmax(td1,d29,in29)"

"l3 = td1/4"    ;NBL = 2, data from two experiments are stored in one file!

"DELTA1=1s/(2 * (cnst6 + 0.07 * (cnst7-cnst6) ) )"
"DELTA3=1s/(2 * (cnst7 - 0.07 * (cnst7-cnst6) ) )"
"DELTA4 = d16 + p16"
"DELTA2=(DELTA1+DELTA3)/2-DELTA4"
"DELTA5=DELTA2-p30/2"
"DELTA8=DELTA2+p30/2-8u"

"DELTA21=1s/(2 * cnst6)-p16-d16"
"DELTA22=1s/(2 * cnst7)-p16-d16"
"DELTA23=d6-p16-d16-4u"
"DELTA24=p2+d0*2"

"d2=1s/(cnst2*2)"      ;cnst2 = 145

1 ze
2 d11

3 d1 do:f2 st0               ;NBL = 2 

  (p1*0.5 ph1):f1            ;HMBC starts here with BANGO using 45 degree 1H pulse 
  d2 pl2:f2
  (center (p2 ph1) (p4 ph1):f2 )
  d2
  (p1*0.5 ph1):f1
  
  DELTA21 UNBLKGRAD
  p16:gp23             ;gpz23 = 15
  d16 pl2:f2
  (p3 ph23):f2
  DELTA22
  p16:gp14            ;gpz14 = -10
  d16
  (p3 ph23):f2
  4u
  p16:gp15            ;gpz15 = -5
  d16
  DELTA23
  (p3 ph24):f2
  d0
  p2 ph22
  d0
  p16:gp1*EA3        ;gpz1 = 80
  d16
  (p24:sp7 ph25):f2  ;p24 = 2msec for Crp80comp.4; 
                     ;Composite 20% smoothed CHIRP adiabatic pulse of 2 ms 
                     ;with 80 kHz sweep 
  DELTA24
  p16:gp1*EA4
  d16 pl2:f2
  (p3 ph25):f2
  4u BLKGRAD

  goscnp ph30

;second experiment starts here: H2OBC

22 d11 st            ;
23 d11 do:f2         

d11 pl12:f2
50u UNBLKGRAD

p16:gp12             ;weak purging gradient between the two experiments 
d16

20u cpd2:f2
(p1 ph1)
d22                 ;d22 decremented
TAU do:f2
4u pl2:f2
(p3 ph3):f2
TAU pl0:f2
(p14:sp30 ph1):f2   ;adiabatic CA-WURST inversion pulse
                    ;sp30:wvm:ad180Cj2: cawurst-20(300 ppm, 0.5 ms; L2H) 
d0
(p2 ph1)
d0
p16:gp1*EA1
TAU8
d25
4u
(p14:sp30 ph1):f2
4u pl2:f2
(p3 ph5):f2

p16:gp1*EA1*-1
TAU4 pl12:f2
d29 cpd2:f2        ;d29 decremented
4u do:f2
4u pl2:f2

(p1 ph2) (p3 ph1):f2

p16:gp1*0.5 
d16 
DELTA5 pl0:f2 

                    ;adiabatic CA-WURST inversion pulse
(p30:sp17 ph1):f2   ;sp17:wvm:ad180Cref: cawurst-20(300 ppm, 0.5 ms)
(p2 ph1) 

p16:gp1*0.75 
d16 
DELTA8 pl12:f2 
4u cpd2:f2 
4u BLKGRAD

go=2 ph31 
                            
d11 do:f2 wr #0 if #0 zd 
   
d11*0.33 igrad EA1
d11*0.33 igrad EA3
d11*0.33 igrad EA4

lo to 2 times 2

d11*0.15 id0 
d11*0.15 dd22 
d11*0.15 dd29 

d11*0.15 ip3*2 

d11*0.15 ip24*2
d11*0.15 ip30*2
d11*0.15 ip31*2

lo to 2 times l3    ; l3 = td1/4   NBL = 2 !

exit

ph1=0 
ph2=1 

ph3=0 2 2 0 
ph5=0 0 2 2 

ph6=0 
ph22=0 
ph23=0 

ph24=0 2 2 0
ph25=0 0 2 2

ph30=0 2 
ph31=0 2 

;pl1 : f1 channel - power level for pulse (default) 
;pl2 : f2 channel - power level for pulse (default) 
;pl12: f2 channel - power level for CPD/BB decoupling 
;p1 : f1 channel - 90 degree high power pulse 

;p2 : f1 channel - 180 degree high power pulse 
;p3 : f2 channel - 90 degree high power pulse 
;p16: homospoil/gradient pulse [1 msec] 
;p24: f2 channel - 180 degree shaped pulse for refocussing 
; = 2msec for Crp80comp.4 
;d0 : incremented delay (2D) [3 usec] 
;d1 : relaxation delay; 1-5 * T1 
;d11: delay for disk I/O [30 msec] 
;d16: delay for homospoil/gradient recovery 
;d22: decremented delay (2D)
;d21: constant time (evolution of J(HH)) [16 - 23 msec] 
;d29: decremented delay (2D)
;cnst6: = 1J(XH)min 
;cnst7: = 1J(XH)max 
;inf1: 1/SW(X) = 2 * DW(X) 
;in0: 1/(2 * SW(X)) = DW(X) 
;in22: = in0 
;in29: = in0 
;nd0: 2 
;ns: 2 * n 
;ds: 16 
;td1: number of experiments 
;FnMODE: echo-antiecho
;NBL = 2

;cpd2: decoupling according to sequence defined by cpdprg2 
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence 

; ~~~~~~~~~~~~~~ WaveMaker Shapes ~~~~~~~~~~~~~~~~~~~~~~~~~~ 
;USE: wvm  command in TopSpin command line to generate CA-WURST adiabatic shape pulses
;sp30:wvm:ad180Cj2: cawurst-20(300 ppm, 0.5 ms; L2H) 
;sp17:wvm:ad180Cref: cawurst-20(300 ppm, 0.5 ms)

;for z-only gradients: 

;use gradient files: 
;gpnam…: SMSQ10.100 

;gpz1 = 80%
;gpz12 = 13% purging gradient


